<commit_before>PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX isvz:    <http://linked.opendata.cz/ontology/isvz.cz/>
PREFIX rdf:     <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

DELETE {
  ?notice ?p ?o .
}
INSERT {
  ?contract ?p ?o .
}
WHERE {
  {
    {
      SELECT ?contract ?notice
      WHERE {
        {
          SELECT ?contract (SAMPLE(?_notice) AS ?notice)
          WHERE {
            ?contract isvz:notice ?_notice .
          }
          GROUP BY ?contract
          HAVING (COUNT(DISTINCT ?_notice) = 1)
          ORDER BY ?contract
        }
      }
      LIMIT {{limit}}
      OFFSET {{offset}}
    }
  }
  ?notice ?p ?o .
  FILTER (?p NOT IN (dcterms:dateSubmitted, dcterms:issued, dcterms:type, rdf:type))
}<commit_msg>Improve stopping condition<commit_after>PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX isvz:    <http://linked.opendata.cz/ontology/isvz.cz/>
PREFIX rdf:     <http://www.w3.org/1999/02/22-rdf-syntax-ns#>

DELETE {
  ?notice ?p ?o .
}
INSERT {
  ?contract ?p ?o .
}
WHERE {
  {
    {
      SELECT ?contract ?notice
      WHERE {
        {
          SELECT ?contract (SAMPLE(?_notice) AS ?notice)
          WHERE {
            ?contract isvz:notice ?_notice .
            ?_notice ?p [] .
            FILTER (?p NOT IN (dcterms:dateSubmitted, dcterms:issued, dcterms:type, rdf:type))
          }
          GROUP BY ?contract
          HAVING (COUNT(DISTINCT ?_notice) = 1)
          ORDER BY ?contract
        }
      }
      LIMIT {{limit}}
    }
  }
  ?notice ?p ?o .
  FILTER (?p NOT IN (dcterms:dateSubmitted, dcterms:issued, dcterms:type, rdf:type))
}
{'fraction_non_alphanumeric': 0.14410480349344978, 'fraction_numerical': 0.012476606363069246, 'mean_word_length': 2.0150375939849625, 'pattern_counts': {'":': 0, '<': 9, '<?xml version=': 0, '>': 9, 'https://': 0, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6074633', 'n_tokens_mistral': 631, 'n_tokens_neox': 586, 'n_words': 122}